Не получается полноценно присоеденить фото к номенклатуре. Конфигурация Розница 2.2 Следующий код:

```1c
    ИмяВременногоФайла="c:\temp\minion.jpg";

    номвл=Справочники.Номенклатура.НайтиПоКоду("1534 ");
    ном=номвл.ПолучитьОбъект();

    КартинкаОбъект = Справочники.НоменклатураПрисоединенныеФайлы.СоздатьЭлемент();
    КартинкаОбъект.ТипХраненияФайла = Перечисления.ТипыХраненияФайлов.ВТомахНаДиске;
    КартинкаОбъект.Наименование = "111-3";
    КартинкаОбъект.ВладелецФайла = номвл;
    КартинкаОбъект.ДатаСоздания=ТекущаяДата();
    К1 = Новый Картинка(ИмяВременногоФайла, истина);
    КартинкаОбъект.ФайлХранилище = Новый ХранилищеЗначения(К1.ПолучитьДвоичныеДанные());
    КартинкаОбъект.Расширение = "jpg";
    КартинкаОбъект.СтатусИзвлеченияТекста = Перечисления.СтатусыИзвлеченияТекстаФайлов.НеИзвлечен;
    ФайлК = Новый Файл(ИмяВременногоФайла);
    КартинкаОбъект.Размер = ФайлК.Размер();
    КартинкаОбъект.Записать();

    ном.Записать();
```

выдает ошибку вида, когда щелкаю по присоединенному файлу в номенклатуре:
---
Вот так получилось:

```1c
    ИмяВременногоФайла="c:\temp\minion.jpg";
    ном=Справочники.Номенклатура.НайтиПоКоду("1536 ");

    ПараметрыФайла=Новый Структура;
    ПараметрыФайла.Вставить("Автор",Пользователи.НайтиПоИмени("Администратор"));
    ПараметрыФайла.Вставить("ВладелецФайлов",ном.Ссылка);
    ПараметрыФайла.Вставить("ИмяБезРасширения","бла-бла-3");
    ПараметрыФайла.Вставить("РасширениеБезТочки","jpg");
    ПараметрыФайла.Вставить("ВремяИзмененияУниверсальное",ТекущаяУниверсальнаяДата());

    ДвоичДанные = Новый ДвоичныеДанные(ИмяВременногоФайла);
    АдресВременногоХранилищаФайла = ПоместитьВоВременноеХранилище(ДвоичДанные);

    фс=ПрисоединенныеФайлы.ДобавитьПрисоединенныйФайл(ПараметрыФайла,АдресВременногоХранилищаФайла,,,);

    об=ном.ПолучитьОбъект();
    об.ФайлКартинки= фс;
    об.Записать();
```
---
изменено по отношению к программе место расположения картинки, загружали с одного, а открываете на другом.
это может быть то, что папка с картинками не расшарена на источнике.
---
Вот это

```1c
    КартинкаОбъект.ТипХраненияФайла = Перечисления.ТипыХраненияФайлов.ВТомахНаДиске;
```

заставляет конфигурацию искать файл на диске, а вы его пихаете в справочник.
Либо ТипХраненияФайла поменяйте, либо положите файл в нужное место и заполните том и путь в томе.

А вот это

```1c
    ном=номвл.ПолучитьОбъект();

    ном.Записать();
```

вообще лишнее
---
Если изменяю тип хранения на "ВИнформационнойБазе", ничего не меняется ((

```1c
    ИмяВременногоФайла="c:\temp\minion.jpg";
    номвл=Справочники.Номенклатура.НайтиПоКоду("1534 ");

    КартинкаОбъект = Справочники.НоменклатураПрисоединенныеФайлы.СоздатьЭлемент();
    КартинкаОбъект.ТипХраненияФайла = Перечисления.ТипыХраненияФайлов.ВИнформационнойБазе;
    КартинкаОбъект.Наименование = "3333";
    КартинкаОбъект.ВладелецФайла = номвл;
    КартинкаОбъект.ДатаСоздания=ТекущаяДата();
    К1 = Новый Картинка(ИмяВременногоФайла, истина);
    КартинкаОбъект.ФайлХранилище = Новый ХранилищеЗначения(К1.ПолучитьДвоичныеДанные());
    КартинкаОбъект.Расширение = "jpg";
    КартинкаОбъект.СтатусИзвлеченияТекста = Перечисления.СтатусыИзвлеченияТекстаФайлов.НеИзвлечен;
    ФайлК = Новый Файл(ИмяВременногоФайла);
    КартинкаОбъект.Размер = ФайлК.Размер();
    КартинкаОбъект.Записать();
```
---
Вот так получилось:

```1c
    ИмяВременногоФайла="c:\temp\minion.jpg";
    ном=Справочники.Номенклатура.НайтиПоКоду("1536 ");

    ПараметрыФайла=Новый Структура;
    ПараметрыФайла.Вставить("Автор",Пользователи.НайтиПоИмени("Администратор"));
    ПараметрыФайла.Вставить("ВладелецФайлов",ном.Ссылка);
    ПараметрыФайла.Вставить("ИмяБезРасширения","бла-бла-3");
    ПараметрыФайла.Вставить("РасширениеБезТочки","jpg");
    ПараметрыФайла.Вставить("ВремяИзмененияУниверсальное",ТекущаяУниверсальнаяДата());

    ДвоичДанные = Новый ДвоичныеДанные(ИмяВременногоФайла);
    АдресВременногоХранилищаФайла = ПоместитьВоВременноеХранилище(ДвоичДанные);

    фс=ПрисоединенныеФайлы.ДобавитьПрисоединенныйФайл(ПараметрыФайла,АдресВременногоХранилищаФайла,,,);

    об=ном.ПолучитьОбъект();
    об.ФайлКартинки= фс;
    об.Записать();
```